
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Kalahasti, Narendra

Narendra.Kalahasti@anritsu.com
2 Corrections required
2.1 Change 1

	Function name
	fl_TC_8_1_4_1_2_TestBody

	Reason for change
	1. According to 38.523-1 Table 8.1.4.1.2.3.3-1, RRC Reconfiguration message is with condition MEAS.
2. Correction for TTCN logging.

3. Changes according to MCC160 comments for R5s201106. 6th parameter for function f_NR_GenerateMeasurementConfigNR should be the ServCellId.

4. Changes similar with R5s200819. RegistrationRequest for Step 16 after the HO may be transmitted immediately after the ReconfigurationComplete which may collide with the HO clean-up procedures on the SS side.



	Summary of change
	1. Use f_NR_SendRRCReconfiguration_MeasNR instead of f_NR_SendRRCReconfiguration.
2. Corrected the TTCN logging

3. As nr_Cell3 is the serving cell during Steps 7-8, add nr_Cell3 for the 6th parameter of f_NR_GenerateMeasurementConfigNR.
4. Apply changes similar to R5s200819.

	TTCN module
	RRC_IntraNR_Handover_NR5GC

	MCC160 Comment
	1. Accepted.
2. Accepted.

3. Accepted.

4. Accepted.


Before change:
    function fl_TC_8_1_4_1_2_TestBody() runs on NR5GC_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;

    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;

    var integer v_Rsrp;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;

    var SIB1 v_SIB1;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    //Power levels of NR cell 1 and cell 3 FR1 acc. Table 8.1.4.1.2.3.2-1, Table 8.1.4.1.2.3.2-2 FFS

    /*         |   Cell 1   |  Cell 3

     * -------------------------------

     *   T0    |    -85     |   -91

     * -------------------------------

     *   T1    |    -85     |   -79

     * -------------------------------

     *   T2    |    -79     |   -85

     * -------------------------------

     *   T3    |    -88     |   -78    *

     */

    v_CellPower_Serving_FR1 := -85;

    v_CellPower_Leaving_FR1 := -91;

    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell3, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, -88, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, -78, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS @sic R5-201477 sic@

    //Set power level to T0

    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1))

    {

       v_Rsrp := 6;

    }

    else

    {

       v_Rsrp := 6;    //FFS dummy  value FR2

    }

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell3), cs_MeasObjectId2(nr_Cell1)};  //@sic R5s200309 sic@

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, -, cs_38508_MeasGapConfig_UE(f_NR_GetGapConfig_InterFreqOdd(nr_Cell1, nr_Cell3))); //@sic R5-199013 sic@

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig); // MeasGap is configured in the SS
    //@siclog "Step 3" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T1".

    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 3?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 5-6" siclog
    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync with rach-ConfigDedicated, keySetChangeIndicator set to true and including nas-Container IE to order the UE to perform inter-frequency handover to NR Cell 3

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 3?

    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell1, tsc_NR_BWP_Id);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 0, fl_BuildIntraN1TransparentContainer ()), //@sic R5-202641 sic@

                                        cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(nr_Cell3))));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 7-8" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1), cs_MeasObjectId2(nr_Cell3)};  //@sic R5s200309 sic@

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList);
    f_NR_SendRRCReconfiguration(nr_Cell3,-,-,-,-, v_MeasConfig);
    //@siclog "Step 9" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T2".

    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 10" siclog
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 1?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell3, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 11-12" siclog
    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync without rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 1

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 1?

    //@sic R5s200309 sic@

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell3,

                                        nr_Cell1,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 0, fl_BuildIntraN1TransparentContainer ())); //@sic R5-202641 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Step 13" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T3".

    f_NR_SetCellPowerList (v_CellPowerList_AtT3);

    //@siclog "Steps 14-15" siclog
    //The SS transmits an RRCReconfiguration including dedicatedSIB1-Delivery containing SIB1 of NR Cell 3 to order the UE to perform handover to NR Cell 3

    //Check: Does the UE transmit an RRCReconfigurationComplete message on NR Cell 3?

    v_SIB1 := f_NR_GetSIB1(nr_Cell3);

    v_SIB1.cellAccessRelatedInfo.plmn_IdentityList[0].trackingAreaCode := int2bit(2,24);

    f_NR_CellInfo_SetTAC(nr_Cell3, int2bit(2,24)); //@sic R5s200309 sic@

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        false, // RBConfig_KeyChange := false

                                        -,

                                        v_SIB1); // SIB1
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog
    //Check: Does UE transmit a REGISTRATION REQUEST message to update the registration of the actual tracking area?

    //This message will be received in SRB2, as this SRB was established after the HO

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp; //@sic R5s200309 sic@

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility, ?, -,-,-,-,-,-, *)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");

    }

    //@siclog "Steps 17" siclog
    //SS responds with REGISTRATION ACCEPT message

    //@sic R5s200309 sic@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell3),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell3)),

                                                                               {bit2oct(f_NR_CellInfo_GetTAC(nr_Cell3))}),

                                                   -,-,-,-,-,

                                                   v_GMM_MobilityInfo.SessionStatus

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell3,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 18" siclog@

    //The UE send a REGISTRATION COMPLETE message

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

    }

  }// End of fl_TC_8_1_4_1_2_TestBody
After change:
    function fl_TC_8_1_4_1_2_TestBody() runs on NR5GC_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;

    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;

    var integer v_Rsrp;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;

    var SIB1 v_SIB1;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    //Power levels of NR cell 1 and cell 3 FR1 acc. Table 8.1.4.1.2.3.2-1, Table 8.1.4.1.2.3.2-2 FFS

    /*         |   Cell 1   |  Cell 3

     * -------------------------------

     *   T0    |    -85     |   -91

     * -------------------------------

     *   T1    |    -85     |   -79

     * -------------------------------

     *   T2    |    -79     |   -85

     * -------------------------------

     *   T3    |    -88     |   -78    *

     */

    v_CellPower_Serving_FR1 := -85;

    v_CellPower_Leaving_FR1 := -91;

    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell3, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, -88, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, -78, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS @sic R5-201477 sic@

    //Set power level to T0

    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1))

    {

       v_Rsrp := 6;

    }

    else

    {

       v_Rsrp := 6;    //FFS dummy  value FR2

    }

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell3), cs_MeasObjectId2(nr_Cell1)};  //@sic R5s200309 sic@

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, -, cs_38508_MeasGapConfig_UE(f_NR_GetGapConfig_InterFreqOdd(nr_Cell1, nr_Cell3))); //@sic R5-199013 sic@

    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig, -, tsc_NR_MeasObjectId2);
    //@siclog "Step 3" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T1".

    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 3?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 5-6" siclog@
    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync with rach-ConfigDedicated, keySetChangeIndicator set to true and including nas-Container IE to order the UE to perform inter-frequency handover to NR Cell 3

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 3?

    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell1, tsc_NR_BWP_Id);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 0, fl_BuildIntraN1TransparentContainer ()), //@sic R5-202641 sic@

                                        cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(nr_Cell3))));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 7-8" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1), cs_MeasObjectId2(nr_Cell3)};  //@sic R5s200309 sic@

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, -, -, -, nr_Cell3);
    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell3, v_MeasConfig, -, tsc_NR_MeasObjectId2);
    //@siclog "Step 9" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T2".

    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 10" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 1?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell3, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 11-12" siclog@
    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync without rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 1

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 1?

    //@sic R5s200309 sic@

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell3,

                                        nr_Cell1,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 0, fl_BuildIntraN1TransparentContainer ())); //@sic R5-202641 sic@

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Step 13" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T3".

    f_NR_SetCellPowerList (v_CellPowerList_AtT3);

    //@siclog "Steps 14-15" siclog@
    //The SS transmits an RRCReconfiguration including dedicatedSIB1-Delivery containing SIB1 of NR Cell 3 to order the UE to perform handover to NR Cell 3

    //Check: Does the UE transmit an RRCReconfigurationComplete message on NR Cell 3?

    v_SIB1 := f_NR_GetSIB1(nr_Cell3);

    v_SIB1.cellAccessRelatedInfo.plmn_IdentityList[0].trackingAreaCode := int2bit(2,24);

    f_NR_CellInfo_SetTAC(nr_Cell3, int2bit(2,24)); //@sic R5s200309 sic@

    f_NR5GC_508RRC_IntraNR_HO_InterCell_Steps1_14(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        false, // RBConfig_KeyChange := false

                                        -,

                                        v_SIB1, // SIB1);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog@
    //Check: Does UE transmit a REGISTRATION REQUEST message to update the registration of the actual tracking area?

    //This message will be received in SRB2, as this SRB was established after the HO

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp; //@sic R5s200309 sic@

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility, ?, -,-,-,-,-,-, *)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");

    }

    f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps15_16(nr_Cell1);
    //@siclog "Steps 17" siclog@
    //SS responds with REGISTRATION ACCEPT message

    //@sic R5s200309 sic@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell3),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell3)),

                                                                               {bit2oct(f_NR_CellInfo_GetTAC(nr_Cell3))}),

                                                   -,-,-,-,-,

                                                   v_GMM_MobilityInfo.SessionStatus

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell3,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Steps 18" siclog@

    //The UE send a REGISTRATION COMPLETE message

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

    }

  }// End of fl_TC_8_1_4_1_2_TestBody
2.2 Change 2

	Function name
	fl_BuildIntraN1TransparentContainer

	Reason for change
	As per TS33.501,cl 6.9.2.3.3,

If the target AMF receives the indication of horizontal KAMF derivation (i.e., keyAmfHDerivationInd), it shall derive the

NAS keys from the received KAMF as specified in clause A.8 and set the NAS COUNTs to zero.
After derivation of Kamf, both DL NAS COUNT and UL NAS COUNT shall be set to 0.

	Summary of change
	Set UL NAS COUNT to zero after the KAMF derivation.

	TTCN module
	RRC_IntraNR_Handover_NR5GC

	MCC160 Comment
	Accepted


Before change:
  function fl_BuildIntraN1TransparentContainer ()  runs on NR5GC_PTC return octetstring

  { //@sic R5-202641 sic@

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var NasCountInfo_Type v_NASCountInfo := f_SS_NG_NAS_CountGet(NASCTRL);

    var template (value) IntraN1TransparentContainer v_NASContainer;

    var MessageAuthenticationCode v_CalculatedMac;

    var bitstring v_NASContainerString;

    //Anritsu TTCN CR R5s201076

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    //update KAMF

    v_SecurityParams.KAMF := f_NG_Authentication_A13(true, f_NasSecurity_IncrementCOUNT(v_NASCountInfo.DL),

                                                     v_SecurityParams.KAMF,     

                                                     v_SecurityParams.KDF);

    // Derive KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_NG_Authentication_A8(tsc_NAS_Enc_Alg,

                                                                   v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                   v_SecurityParams.KAMF,

                                                                   v_SecurityParams.KDF);

    // Derive KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_NG_Authentication_A8(tsc_NAS_Int_Alg,

                                                                  v_SecurityParams.NAS_Integrity.Algorithm,

                                                                  v_SecurityParams.KAMF,

                                                                  v_SecurityParams.KDF);

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, omit);

    v_NASContainer := cs_IntraN1TransparentContainer (omit,         // set to omit in order to calculate its own value

                                                      v_SecurityParams.NAS_Ciphering.Algorithm,

                                                      v_SecurityParams.NAS_Integrity.Algorithm,

                                                      '1'B,

                                                      '0'B,

                                                      v_SecurityParams.KSIamf,

                                                      substr(f_NasSecurity_IncrementCOUNT(v_NASCountInfo.DL), 3, 1));

    v_NASContainerString := encvalue(valueof(v_NASContainer));

    v_CalculatedMac := fx_NG_NasIntegrityAlgorithm(bit2oct(v_NASContainerString),

                                                   v_SecurityParams.NAS_Integrity.Algorithm,

                                                   v_SecurityParams.NAS_Integrity.K_NAS,

                                                   'FFFFFFFF'O,  // according to 33.501 cl. 6.9.2.3.3

                                                   '00001'B,

                                                   tsc_DirectionDL);

    v_NASContainer.mac := v_CalculatedMac;

    v_NASContainerString := encvalue(v_NASContainer);

    //Anritsu TTCN CR R5s201076

    f_SS_NG_NAS_CountSet(NASCTRL, '00000000'O, omit);    

    v_RRC_SecurityParams.KAMF := v_SecurityParams.KAMF;

    v_RRC_SecurityParams.KSIamf := v_SecurityParams.KSIamf; 

    // Derive KgNB

    v_RRC_SecurityParams.KGNB := f_NR_Authentication_A9(v_RRC_SecurityParams.KDF, v_RRC_SecurityParams.KAMF, 'FFFFFFFF'O);

    v_RRC_SecurityParams.NH := v_RRC_SecurityParams.KGNB;

    v_RRC_SecurityParams.NCC := 0;

    f_NR_Security_Set(v_RRC_SecurityParams);

    f_NR5GC_Security_Set(v_SecurityParams);   

    return bit2oct(v_NASContainerString);

  }
After change:
  function fl_BuildIntraN1TransparentContainer ()  runs on NR5GC_PTC return octetstring

  { //@sic R5-202641 sic@

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var NasCountInfo_Type v_NASCountInfo := f_SS_NG_NAS_CountGet(NASCTRL);

    var template (value) IntraN1TransparentContainer v_NASContainer;

    var MessageAuthenticationCode v_CalculatedMac;

    var bitstring v_NASContainerString;

    //Anritsu TTCN CR R5s201076

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    //update KAMF

    v_SecurityParams.KAMF := f_NG_Authentication_A13(true, f_NasSecurity_IncrementCOUNT(v_NASCountInfo.DL),

                                                     v_SecurityParams.KAMF,     

                                                     v_SecurityParams.KDF);

    // Derive KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_NG_Authentication_A8(tsc_NAS_Enc_Alg,

                                                                   v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                   v_SecurityParams.KAMF,

                                                                   v_SecurityParams.KDF);

    // Derive KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_NG_Authentication_A8(tsc_NAS_Int_Alg,

                                                                  v_SecurityParams.NAS_Integrity.Algorithm,

                                                                  v_SecurityParams.KAMF,

                                                                  v_SecurityParams.KDF);

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering, omit);

    v_NASContainer := cs_IntraN1TransparentContainer (omit,         // set to omit in order to calculate its own value

                                                      v_SecurityParams.NAS_Ciphering.Algorithm,

                                                      v_SecurityParams.NAS_Integrity.Algorithm,

                                                      '1'B,

                                                      '0'B,

                                                      v_SecurityParams.KSIamf,

                                                      substr(f_NasSecurity_IncrementCOUNT(v_NASCountInfo.DL), 3, 1));

    v_NASContainerString := encvalue(valueof(v_NASContainer));

    v_CalculatedMac := fx_NG_NasIntegrityAlgorithm(bit2oct(v_NASContainerString),

                                                   v_SecurityParams.NAS_Integrity.Algorithm,

                                                   v_SecurityParams.NAS_Integrity.K_NAS,

                                                   'FFFFFFFF'O,  // according to 33.501 cl. 6.9.2.3.3

                                                   '00001'B,

                                                   tsc_DirectionDL);

    v_NASContainer.mac := v_CalculatedMac;

    v_NASContainerString := encvalue(v_NASContainer);

    //Anritsu TTCN CR R5s201076

    f_SS_NG_NAS_CountSet(NASCTRL, '00000000'O, '00000000'O);

    v_RRC_SecurityParams.KAMF := v_SecurityParams.KAMF;

    v_RRC_SecurityParams.KSIamf := v_SecurityParams.KSIamf; 

    // Derive KgNB

    v_RRC_SecurityParams.KGNB := f_NR_Authentication_A9(v_RRC_SecurityParams.KDF, v_RRC_SecurityParams.KAMF, 'FFFFFFFF'O);

    v_RRC_SecurityParams.NH := v_RRC_SecurityParams.KGNB;

    v_RRC_SecurityParams.NCC := 0;

    f_NR_Security_Set(v_RRC_SecurityParams);

    f_NR5GC_Security_Set(v_SecurityParams);   

    return bit2oct(v_NASContainerString);

  }
